Agilent PN 8566B/8568B-1

Third Order Intermodulation Distortion Measurements: A Downloadable Procedure for Agilent 8566B and 8568B Spectrum Analyzers

Product Note

Introduction

Third order intermodulation distortion (IMD) measurements are complex and can be tedious when performed manually, even when sophisticated measuring instruments such as Agilent Technologies 8566B or 8568B Spectrum Analyzers are used. Performing such measurements automatically from a computer is far more efficient, since less operator time and effort is required. The downloadable capability of an 8566B or 8568B Spectrum Analyzer adds to this efficiency by making it possible to execute programs that are stored in the analyzer and get results, instead of data, from the analyzer.

This product note contains a complete program that measures the third order IMD products of two input signals in the range of 10 MHz to 500 MHz. This program can be downloaded into an 8566B or 8568B Spectrum Analyzer and executed from the analyzer front panel, or run from a computer.

Prerequisites

To fully understand the third order IMD program, some experience in analyzer programming is necessary. The *Introductory Operation Guide* (Agilent publication number 5952-9389) and the *Agilent 8566B* or *8568B Operating and Programming Manual* (Agilent part numbers 08566-90040 and 08568-90041, respectively) are good references for developing an understanding of analyzer programming. An understanding of downloadable programming concepts is also required. *A Structured Approach to Downloadable Programming* (Agilent publication number 5952-9392) is a good source for such information. Other references that may prove useful are *Spectrum Analysis... Distortion Measurements* (Agilent publication number 5952-9235), and *Quick Reference Guide for Agilent 8566B and Agilent 8568B Spectrum Analyzers* (Agilent publication number 5955-8970).

Equipment

A list of the equipment necessary to make third order IMD measurements follows:

1–Agilent 8566B or 8568B Spectrum Analyzer

2–Signal sources, ranging in frequency from 10 MHz to 500 MHz (such as Agilent 8640B and 8656B Signal Generators)

1–Signal combiner, such as an Agilent 8721A Directional Coupler (up to 100 MHz), 11667A power splitter, or Weinschel 1502 Combiner

2–6 dB pads (recommended to improve isolation between signal sources)

2–Low-pass filters (recommended if large IMD products are generated within the signal sources)

Connecting cables, power cords, and adapters where necessary

Agilent Technologies
Innovating the HP Way

Test Setup

To simulate a test for third order IMD, connect the equipment as shown in Figure 1:

Why Measure Third Order Intermodulation Distortion?

Two-tone third order intermodulation is a common problem in narrow-band systems. When two (or more) signals are present in a system, strong harmonic components are often generated (See Figure 2). In cases where two signals are present, the two signals (f_1 and f_2) mix with each other's second harmonic ($2f_1$ and $2f_2$) and create distortion products evenly spaced about the fundamentals ($2f_1–f_2$ and $2f_2–f_1$). Components such as amplifiers, mixers, and filters can generate third order intermodulation distortion products.

These distortion products can degrade the performance of many communication systems, such as FM and AM transceivers and high frequency radio teletypes. For example, signals transmitted with excessive third order IMD can interfere with other transmissions. Receivers must also be distortion-free, especially in the preamplifier stages, to prevent crosstalk between adjacent channels.

Third Order Intermodulation Distortion Program

The program contained in this product note is designed to test for two-tone third order intermodulation distortion. The program begins by bringing the test tones on the analyzer display, setting the attenuator for optimum dynamic range, and measuring the amplitudes and frequencies of the input signals. It then locates the third order IMD products, measures them, and returns their amplitudes in dBc. It also reports the frequency separation and level of the test tones. If the third order IMD products are too low in amplitude to be measured, it reports this instead. The program can be executed from a computer or from the front panel of the analyzer.

The input signals must be in the range of 10 MHz to 500 MHz. They can be separated anywhere from 1 kHz to 10 kHz, and the difference in their amplitudes should be less than 2 dB. The program takes approximately 22 seconds to execute when measuring test tones that are 3 kHz to 10 kHz apart; it takes approximately 94 seconds to execute when measuring test tones that are 1 kHz and 3 kHz apart. The program is accurate to within ±0.4 dB for signal separations from 3 kHz to 10 kHz, with the spectrum analyzer operating in the corrected* mode. For signal separations from 1 kHz to 3 kHz, accuracy is ±1.5 dB, with the analyzer operating in the corrected* mode.

When typing the program into the computer, follow all Agilent 8566B and 8568B programming syntax requirements. By typing the program as shown, these requirements will be met. Program comments (text appearing on a line after an exclamation point) may be omitted.

If downloadable procedures are already stored in the analyzer's RAM, it may be necessary to dispose of some or all of these procedures to provide the 4802 bytes of memory space required for the third order IMD program. To dispose of all procedures stored in RAM, type:

```
OUTPUT 718;"DISPOSE ALL;"
```

and press **[EXECUTE]**.

* To access the corrected mode, press **[SHIFT]** **[W]**, and **[SHIFT]** **[X]**. The analyzer must be operating in the 20 °C to 30 °C temperature range. The accuracy of the results is reduced when the analyzer is not operated in the corrected mode.



Figure 2. Two-Tone Third Order Intermodulation (Fundamentals and IMD Products)



Figure 1. Test Setup for Third Order IMD Measurements

Program List

The following is a line-by-line listing of the program, followed by the program annotation.

```
10    ! Filename:  T_HIRDIMOD          Date:  25.12.84
20    ! Description of program:        This is a downloadable program for the
30    !                                8566B and 8568B that measures the
40    !                                third order intermodulation products of
50    !                                two equal amplitude signals that are
60    !                                separated in frequency from 1 kHz to
70    !                                10 kHz.  The test tones must be within
80    !                                the 10 MHz to 500 MHz frequency range.
90    !                                The program sets the attenuator for opti-
100   !                                mum dynamic range and tests that the test
110   !                                tones are equal in amplitude (within 2 dB).
120   !                                The program reports if no third order
130   !                                products appear above the noise level.
140   ! Executable using:              T_HIRDIMOD or <SHIFT> <2> <Hz>
150   ! Execution time:                Separations from 3 kHz to 10 kHz; 22 sec.
160   !                                Separations from 1 kHz to 3 kHz;  94 sec.
170   !
180   ! Required memory allocation:  4802 bytes
190   !
200   ! Program begins here. . .
210   !
220   ASSIGN @Sa TO 718
230   !
240   ! INITIALIZE VARIABLES AND TRACES
250   !
260   OUTPUT @Sa;"VARDEF O_PTRANGE,0;VARDEF T_ONE,0;VARDEF T_TWO,0;"
270   OUTPUT @Sa;"VARDEF T_ONEAMP,0;VARDEF T_TWOAMP,0;VARDEF H_EIGHT,0;"
280   OUTPUT @Sa;"VARDEF D_IFF,0;VARDEF N_OISE,0;"
290   OUTPUT @Sa;"VARDEF I_MRDBC,0;VARDEF I_MLDBC,0;"
300   OUTPUT @Sa;"VARDEF H_OLD,0;VARDEF L_EFT,0;"
310   OUTPUT @Sa;"TRDEF S_TORE,1008;"
320   !
330   ! FIND THE TEST TONES AND STORE THEIR FREQUENCY AND AMPLITUDE
340   ! IN PREDEFINED VARIABLES
350   !
360   OUTPUT @Sa;"FUNCDEF T_ESTTONES, @"
370   OUTPUT @Sa;"IP;SNGLS;EM;FA 10 MZ;FB 500 MZ;"
380   OUTPUT @Sa;"TS;MKPK HI;MKTRACK ON;"
390   OUTPUT @Sa;"SP30KZ;VB1KZ;TS;"
400   OUTPUT @Sa;"MKTRACK OFF;TS;MKPK HI;"
410   OUTPUT @Sa;"IF MA,GT,RL THEN;"
420   OUTPUT @Sa;"  REPEAT;"
430   OUTPUT @Sa;"   RL UP;TS;MKPK HI;"
440   OUTPUT @Sa;"  UNTIL MA,LE,RL;"
450   OUTPUT @Sa;"ENDIF;"
460   OUTPUT @Sa;"ADD O_PTRANGE,MA,38;"
470   OUTPUT @Sa;"IF AT,LT,O_PTRANGE THEN;"
480   OUTPUT @Sa;"  REPEAT;"
490   OUTPUT @Sa;"   AT UP;"
500   OUTPUT @Sa;"  UNTIL AT,GE,O_PTRANGE;"
510   OUTPUT @Sa;"ENDIF;"
520   OUTPUT @Sa;"MKRL;TS;"
530   OUTPUT @Sa;"MOV T_ONE,MF;"
540   OUTPUT @Sa;"MOV T_ONEAMP,MA;"
550   OUTPUT @Sa;"MKPX 10DB;"
560   OUTPUT @Sa;"MKPK NH;"
570   OUTPUT @Sa;"MOV T_TWO,MF;"
580   OUTPUT @Sa;"MOV T_TWOAMP,MA;"
590   OUTPUT @Sa;"@;"
600   !
610   ! IF THE TEST TONES ARE NOT OF EQUAL AMPLITUDE
```

Program Annotation

Lines 10–200	Program name and description
Line 220	Assign the spectrum analyzer address, 718, as **@Sa**.
Lines 260–310	Define the variables in the program and set their initial values.
Line 360	Define the function in Lines 370–590 as **T_ESTTONES**.
Lines 370–590	Set the spectrum analyzer's start frequency to 10 MHz and stop frequency to 500 MHz. Zoom in on the higher amplitude test tone to a 30 kHz span. If the peak of the signal is above the reference level, increase the reference level until the signal peak is at or below it. Set the attenuator for optimum dynamic range, then move the signal peak to the reference level and measure it. Store the signal's frequency value in variable **T_ONE** and its amplitude value in variable **T_ONEAMP**. Locate the second test tone, measure it, and store its frequency and amplitude values in variables **T_TWO** and **T_TWOAMP**, respectively.

```
620   ! OPERATOR IS PROMPTED TO ADJUST THEM WITHIN 2 dB
630   ! OF EACH OTHER
640   !
650   OUTPUT @Sa;"FUNCDEF E_QUALAMP, @"
660   OUTPUT @Sa;"SUB H_EIGHT,T_ONEAMP,T_TWOAMP;"
670   OUTPUT @Sa;"IF H_EIGHT,LT,0;"
680   OUTPUT @Sa;"  THEN SUB H_EIGHT,0,H_EIGHT;"
690   OUTPUT @Sa;"ENDIF;"
700   OUTPUT @Sa;"IF H_EIGHT,GT,2 THEN;"
710   OUTPUT @Sa;"  CONTS;DA3072;D3;PU;PA100,600;TEXT /ADJUST TEST TONES FOR
EQUAL/;HD;"
720   OUTPUT @Sa;"  PU;PA100,550;TEXT /AMPLITUDE AND PRESS THE HZ KEY/;"
730   OUTPUT @Sa;"  SS EP;"
740   OUTPUT @Sa;"  EM;SNGLS;TS;MKPK HI;"
750   OUTPUT @Sa;"  MOV T_ONE,MF;"
760   OUTPUT @Sa;"  MOV T_ONEAMP,MA;"
770   OUTPUT @Sa;"  MKPK NH;"
780   OUTPUT @Sa;"  MOV T_TWO,MF;"
790   OUTPUT @Sa;"  MOV T_TWOAMP,MA;"
800   OUTPUT @Sa;"ENDIF;"
810   OUTPUT @Sa;"@;"
820   !
830   ! FIND THE THIRD ORDER PRODUCTS
840   !
850   OUTPUT @Sa;"FUNCDEF P_RODUCTS, @"
860   OUTPUT @Sa;"IF T_ONE,GE,T_TWO THEN ;"
870   OUTPUT @Sa;"  XCH T_ONE,T_TWO;"
880   OUTPUT @Sa;"  XCH T_ONEAMP,T_TWOAMP;"
890   OUTPUT @Sa;"ENDIF;"
900   OUTPUT @Sa;"SUB D_IFF,T_TWO,T_ONE;"
910   OUTPUT @Sa;"DIV H_OLD,D_IFF,2;"
920   OUTPUT @Sa;"ADD CF,T_ONE,H_OLD;"
930   OUTPUT @Sa;"IF D_IFF,LT,3000 THEN;"
940   OUTPUT @Sa;"  SP DN;"
950   OUTPUT @Sa;"ENDIF;"
960   OUTPUT @Sa;"TS;MOV S_TORE,TRA;"
970   OUTPUT @Sa;"SAVES 2;"
980   OUTPUT @Sa;"@;"
990   !
1000  ! MEASURE THIRD ORDER PRODUCTS
1010  !
1020  OUTPUT @Sa;"FUNCDEF M_EASURE, @"
1030  OUTPUT @Sa;"ADD CF,T_TWO,D_IFF;"
1040  OUTPUT @Sa;"SP;DN;DN;TS;"
1050  OUTPUT @Sa;"IF D_IFF,GE,3000 THEN;"
1060  OUTPUT @Sa;"  MKPK HI;MKRL;MOV VB,RB;"
1070  OUTPUT @Sa;"  VB;DN;TS;MKPK HI;"
1080  OUTPUT @Sa;"ELSE MKN;SP;DN;TS;MKPK HI;"
1090  OUTPUT @Sa;"  MKRL;VB;DN;TS;MKPK HI;"
1100  OUTPUT @Sa;"ENDIF;"
```

Line 650	Define the function in Lines 660–810 as **E_QUALAMP**.
Lines 660–810	Measure the difference in amplitude between the test tones. If the difference is less than 0 dB, take the absolute value of the difference. If this absolute value is less than 2 dB, continue to the next function. If it is greater than 2 dB, instruct the operator to adjust the test tones for equal amplitude. Re-measure the frequency and amplitude of the test tones, store the new frequency and amplitude values in variables **T_ONE, T_ONEAMP, T_TWO** and **T_TWOAMP.**
Line 850	Define the function in Lines 860–980 as **P_RODUCTS.**
Lines 860–890	If the frequency and amplitude values of the tone higher in frequency are stored in variables **T_ONE** and **T_ONEAMP**, exchange these values with the frequency and amplitude values stored in **T_TWO** and **T_TWOAMP**, respectively. This assures that variable **T_TWO** stores the frequency value of the tone higher in frequency.
Lines 900–980	Measure the difference in frequency between the test tones, store the difference in variable **D_IFF**, and set the center frequency to halfway between the two test tones (**T_ONE** + ½ **D_IFF**). If the value of **D_IFF** is less than 3 kHz, also reduce the frequency span. Store the resulting trace in **S_TORE** and the instrument settings in Register 2.
Line 1020	Define the function in Lines 1030–1160 as **M_EASURE**.
Lines 1030–1040	Add the values of variables **T_TWO** and **D_IFF**, and set the center frequency to the sum. This sum is equal to the frequency of the upper third order IMD product, which appears on the right side of the display.
Lines 1050–1100	If the tones are separated by more than 3 kHz, set the third order IMD product to the reference level and measure its amplitude. If the tone separation is less than 3 kHz, reduce the frequency span. This will also reduce the resolution bandwidth. (Shape factor constraints necessitate the narrower bandwidth, and the reduced scan prevents a lengthy sweeptime.) Set the third order IMD product to the reference level and measure its amplitude.

```
1110  OUTPUT @Sa;"SUB I_MRDBC,MA,T_TWOAMP;"
1120  OUTPUT @Sa;"SUB CF,T_ONE,D_IFF;"
1130  OUTPUT @Sa;"TS;MKPK HI;"
1140  OUTPUT @Sa;"MOV L_EFT,MA;"
1150  OUTPUT @Sa;"SUB I_MLDBC,MA,T_ONEAMP;"
1160  OUTPUT @Sa;"@;"
1170  !
1180  ! REPORT RESULTS ON SCREEN
1190  !
1200  OUTPUT @Sa;"FUNCDEF R_EPORT, @"
1210  OUTPUT @Sa;"VIEW TRA;RCLS 2;MOV TRA,S_TORE;"
1220  OUTPUT @Sa;"DA3072;D2;PU;PA300,800;TEXT /INTERMODULATION PRODUCTS/
1230  OUTPUT @Sa;"PU;PA200,750;TEXT /TEST TONE LEVEL =      /;DSPLY
T_ONEAMP,5.2;PU;PA 700,750;TEXT /dBm/;"
1240  OUTPUT @Sa;"PU;PA200,700;TEXT /TEST TONE SEPARATION = /;DSPLY
D_IFF,6.0;PU;PA700,700;TEXT /Hz/;"
1250  OUTPUT @Sa;"PU;PA200,630;TEXT /THIRD ORDER/;"
1260  OUTPUT @Sa;"PU;PA200,550;TEXT /LOWER/;PU;PA 564,550;DSPLY
I_MLDBC,5.2;PU;PA700,550;TEXT /dBc/;"
1270  OUTPUT @Sa;"PU;PA200,500;TEXT /UPPER/;PUPA 564,500;DSPLY
I_MRDBC,5.2;PU;PA700,500;TEXT /dBc/;"
1280  OUTPUT @Sa;"PU;PA300,82;TEXT /Press SHIFT 2 Hz to repeat test/;HD;
1290  OUTPUT @Sa;"@;"
1300  !
1310  ! THE FOLLOWING REPORTS THAT NO THIRD ORDER
1320  ! PRODUCTS ARE FOUND ABOVE THE NOISE LEVEL
1330  !
1340  OUTPUT @Sa;"FUNCDEF N_OTHIRD, @"
1350  OUTPUT @Sa;"RCLS 2;MOV TRA,S_TORE;"
1360  OUTPUT @Sa;"EM;D3;DA3072;PU;PA100,600;"
1370  OUTPUT @Sa;"TEXT /THIRD ORDER INTERMODULATION PRODUCTS/;"
1380  OUTPUT @Sa;"PU;PA100,550;TEXT /ARE AT OR BELOW THE NOISE LEVEL/;"
1390  OUTPUT @Sa;"PU;PA100,525;TEXT /Press SHIFT 2 Hz to repeat test/;"
1400  OUTPUT @Sa;"@;"
1410  !
1420  ! IF NO THIRD ORDER PRODUCTS APPEAR
1430  ! ABOVE THE NOISE LEVEL, THE FOLLOWING
1440  ! BRANCHES THE PROGRAM TO READ OUT
1450  ! THAT THE INTERMOD PRODUCTS ARE
1460  ! AT OR BELOW THE NOISE LEVEL
1470  ! OTHERWISE IT REPORTS THE MEASURED
1480  ! RESULTS.
1490  !
1500  OUTPUT @Sa;"FUNCDEF C_HECK, @"
1510  OUTPUT @Sa;"SMOOTH TRA,32;MKMIN;"
1520  OUTPUT @Sa;"MOV N_OISE,MA;ADD N_OISE,N_OISE,15;"
1530  OUTPUT @Sa;"IF L_EFT,LE,N_OISE THEN;"
1540  OUTPUT @Sa;"  N_OTHIRD;"
1550  OUTPUT @Sa;"  ELSE R_EPORT;"
1560  OUTPUT @Sa;"ENDIF;"
1570  OUTPUT @Sa;"@;"
```

Line 1110	Measure the difference between the value in variable **T_TWOAMP** and the amplitude of the upper third order IMD product, and store the result in variable **I_MRDBC**. This result is the relative amplitude of the third order IMD product, in dBc from the peak of the higher frequency tone.
Lines 1120–1160	Subtract the values of variables **T_ONE** and **D_IFF**, and set the center frequency to the difference. This difference is equal to the frequency of the lower third order IMD product, which appears on the left side of the display. Measure the amplitude of the IMD product, and store it in variable **L_EFT**. Measure the difference between the value in variable **T_ONEAMP** and the amplitude of the third order IMD product, and store the result in variable **I_MLDBC**. This result is the relative amplitude of the third order IMD product, in dBc from the peak of the lower frequency tone.
Line 1200	Define the function in Lines 1210–1290 as **R_EPORT**.
Lines 1210–1290	Display the user-defined trace, **S_TORE**, the settings stored in Register 2, the test tone amplitude and separation, and the relative amplitude values (in dBc) of the upper and lower third order IMD products. Prompt the user to press shift 2 Hz to re-run the test.
Line 1340	Define the function in Lines 1350–1400 as **N_OTHIRD**.
Lines 1350–1400	Display the user-defined trace, **S_TORE**, and the settings stored in Register 2. Report the third order IMD products are at or below the noise level.
Line 1500	Define the function in Lines 1510–1570 as **C_HECK**.
Lines 1510–1570	Smooth the trace. The minimum level of the smoothed trace is an approximation of the average noise level. Compare this minimum trace level to the amplitude of the lower third order IMD product, stored in variable **L_EFT**. If **L_EFT** is 15 dB higher than the approximate noise level, branch to **R_EPORT**. If **L_EFT** is not 15 dB higher than the approximate noise level, branch to **N_OTHIRD**.

```
1580  !
1590  ! MAIN FUNCTION
1600  !
1610  OUTPUT @Sa;"FUNCDEF T_HIRDIMOD, @"
1620  OUTPUT @Sa;"T_ESTTONES;E_QUALAMP;P_RODUCTS;M_EASURE;C_HECK;"
1630  OUTPUT @Sa;"@;"
1640  !
1650  ! DEFINE THE PROGRAM AS SOFTKEY TWO
1660  !
1670  OUTPUT @Sa;"KEYDEF 2,T_HIRDIMOD;"
1680  END
```

Line 1610	Define the main function in Line 1620 as **T_HIRDIMOD**.
Line 1620	Specify the sequence of the functions to test for third order IMD products: find the test tones, check for equal amplitude, locate and measure the third order IMD products, and ensure the products are above the noise level. If the products are above the noise level, report the measured results, or else report they are too low to be measured.
Line 1670	Make the program accessible from the front panel under softkey 2.

Program Execution

To execute the program from the analyzer front panel, press **[SHIFT][2][Hz]**. Or, to execute it from a controller, type:

```
OUTPUT 718;"T_HIRDIMOD;"
```

and press **[EXECUTE]**.

Application Example

To measure the internal third order IMD of an amplifier:

Before testing an amplifier (or any other device), run the program without the amplifier in the system to ensure the system is not producing intermodulation distortion products. If the program results show the products are above the noise, increase the attenuation at the output of the signal sources or check the low-pass filters to ensure they are not passing any test-tone harmonics. When the system distortion is eliminated, insert the amplifier between the output of the combiner and the input of the analyzer as shown in the section, Test Setup.



Figure 3. Example of T_HIRDIMOD Results

By internet, phone, or fax, get assistance with all your test and measurement needs.

Online Assistance

www.agilent.com/find/assist

Printed in U.S.A. 10/00
5954-2701

Agilent Technologies

Innovating the HP Way